h dftactgrp(*no)                                               

dModUsage         pr                                           

d QModNam                       20    const                    

d SrchLib                       10    const                    

dModUsage         pi                                           

d QModNam                       20    const                    

d SrchLib                       10    const                    

d/copy qsysinc/qrpglesrc,qgyolobj                              

d/copy qsysinc/qrpglesrc,qgy                                   

d/copy qsysinc/qrpglesrc,qusec                                 

d/copy qsysinc/qrpglesrc,qusgen                                

dChkSrvPgmMod     pr                                           

dChkPgmMod        pr                                             

dOpnLstDone       pr              n                              

d RqsHandle                      4                               

dSndErrMsg        pr                                             

dSetup            pr                                             

dOpnListObj       pr                  extpgm('QGYOLOBJ')         

d RcvVar                         1    options(*varsize)          

d LenRcvVar                     10i 0 const                      

d ListInfo                      80                               

d NbrRcdRqs                     10i 0 const                      

d SortInfo                            const likeds(QGYOSI)       

d QualObjNam                    20    const                      

d ObjType                       10    const                      

d AuthCtl                             const likeds(QGYOAC)       

d SelCtl                              const likeds(SelCtl)       

d NbrKeys                       10i 0 const                      

d KeysToRtn                     10i 0 const                            

d QUSEC                               likeds(QUSEC)                    

d JobID                      65535    const options(*nopass :*varsize) 

d JobIDFmt                       8    const options(*nopass)           

d ASPCtl                     65535    const options(*nopass :*varsize) 

dGetNextEnt       pr                  extpgm('QGYGTLE')                

d RcvVar                         1    options(*varsize)                

d LenRcvVar                     10i 0 const                            

d RqsHandle                      4    const                            

d ListInfo                      80                                     

d NbrRcdRqs                     10i 0 const                            

d StrRcd                        10i 0 const                            

d QUSEC                               likeds(QUSEC)                    

dCloseList        pr                  extpgm('QGYCLST')                

d RqsHandle                      4    const                            

d QUSEC                               likeds(QUSEC)               

dGenHdr           ds                  likeds(QUSH0100)            

d                                     based(ModLstSpcPtr)         

dRcvVarE          ds                  based(RcvVarEPtr)           

d                                     likeds(QGYORV01)            

dSelCtl           ds                  qualified                   

d Ctl                                 likeds(QGYOSC)              

d Status                         1                                

dModToFnd         ds                                              

d ModNam                        10                                

d ModNamLib                     10                                

dObjLibNam        ds                                              

d                               10    inz('*ALL')               

d ObjLib                        10                              

dErrCde           ds                  qualified                 

d Hdr                                 likeds(QUSEC)             

d MsgDta                       128                              

dRcvVar           s           4096                              

dPgmHdl           s              4                              

dSrvPgmHdl        s              4
dOLPgmOpn         s               n   inz(*off) 

dOLSrvPgmOpn      s               n   inz(*off)                               

dRcvVarEPtr       s               *                             

dModLstSpcPtr     s               *                             

dCount            s             10i 0                           

dNbrMods          s             10i 0                           

dHits             s             10i 0                           

dWait             s              1                              

dModLstSpc        c                   'MODLUSRSPCQTEMP'          

 /free                                                           

  Monitor;                                                       

  // check that the required parameters were passed              

  if %parms < 2;                                                 

     dsply 'There are two required parameters:';                 

     dsply '1. The module you are searching for';                

     dsply '2. The library to be searched in';                   

     dsply 'The program did not run.' ' ' Wait;                  

     *inlr = *on;                                                

     return;                                                     

  endif;                                                         

  Setup();                                                       

  // Get the initial set of *SRVPGM list entries                       

  GetNextEnt( RcvVar :%size(RcvVar) :SrvPgmHdl :QGYLI :50 :1 :ErrCde); 

  select;                                                              

     when ErrCde.Hdr.QUSBAVL = 0;           // List is available       

          dow (QGYIC07 = 'C') or (QGYIC07 = 'P');                      

              RcvVarEPtr = %addr(RcvVar);                              

              for Count = 1 to QGYRRTN02;                              

                  ChkSrvPgmMod();                                      

                  RcvVarEPtr += QGYRL07;                               

              endfor;                                                  

              if OpnLstDone(SrvPgmHdl);
                 OLSrvPgmOpn = *off;                                  

                 leave;                                                

              endif;                                                   

          enddo;                                                       

          if QGYIC07 = 'I';                                           

             dsply ('Unable to access all *SRVPGMs in ' + ObjLib);    

          else;                                                       

             dsply (%char(Hits) + ' *SRVPGM instances of ' +          

                    %trimr(ModNam) + ' found.');                      

          endif;                                                      

     when ErrCde.Hdr.QUSEI = 'GUI0006';     // An empty list          

          CloseList( SrvPgmHdl :QUSEC);                       
          OLSrvPgmOpn = *off;                        
          dsply ('No *SRVPGMs found in ' + SrchLib);                  

     other;                                 // Error situation        

          SndErrMsg();                                                

  endsl;                                                              

  Hits = 0;                                                           

  // Get the initial set of *PGM list entries                         

  GetNextEnt( RcvVar :%size(RcvVar) :PgmHdl :QGYLI :50 :1 :ErrCde);     

  select;                                                               

     when ErrCde.Hdr.QUSBAVL = 0;           // List is available        

          dow (QGYIC07 = 'C') or (QGYIC07 = 'P');                       

              RcvVarEPtr = %addr(RcvVar);                               

              for Count = 1 to QGYRRTN02;                               

                  ChkPgmMod();                                          

                  RcvVarEPtr += QGYRL07;                                

              endfor;                                                   

              if OpnLstDone(PgmHdl);                                    

                 OLPgmOpn = *off;                         

                 leave;                                                 

              endif;                                                    

          enddo;                                                        

          if QGYIC07 = 'I';                                             

             dsply ('Unable to access all *PGMs in ' + ObjLib);         

          else;                                                    

             dsply (%char(Hits) + ' *PGM instances of ' +          

                    %trimr(ModNam) + ' found.') ' ' Wait;          

          endif;                                                   

     when ErrCde.Hdr.QUSEI = 'GUI0006';     // An empty list       

          OLPgmOpn = *off;                        

          dsply ('No *PGMs found in ' + SrchLib) ' ' Wait;         

     other;                                 // Error situation     

          SndErrMsg();                                             

  endsl;                                                           

  *inlr = *on;                                                     

  return;                                                          

  on-error;                                                        

  if OLPgmOpn;                                            

     CloseList( PgmHdl :QUSEC);                                       

  endif;                                                              

  if OLSrvPgmOpn;                                            

     CloseList( SrvPgmHdl :QUSEC);                                    

  endif;                                                              

  dsply 'MODUSAGE did not complete. See job log.' ' ' Wait;
  *inlr = *on;                                                        

  return;                                                             

  endmon;                                                             

 /end-free                                                            

 *****************************************************************    

pChkSrvPgmMod     b                                                   

dChkSrvPgmMod     pi                                             

dGetSrvPgmInfo    pr                  extpgm('QBNLSPGM')         

d QUsrSpcNam                    20    const                      

d Format                         8    const                      

d QSrvPgmNam                    20    const                      

d QUSEC                               likeds(QUSEC)              

d/copy qsysinc/qrpglesrc,qbnlspgm                                

dSrvPgmE          ds                  likeds(QBNL010000)         

d                                     based(SrvPgmEPtr)          

dSrvPgmEPtr       s               *                              

 /free                                                           

  // Get the list of bound *MODULEs for current *SRVPGM               

  GetSrvPgmInfo( ModLstSpc :'SPGL0100'                                

                :(RcvVarE.QGYON00 + RcvVarE.QGYOL03) :QUSEC);         

  if (GenHdr.QUSIS = 'C');                                            

     SrvPgmEPtr = ModLstSpcPtr + GenHdr.QUSOLD;                       

     for NbrMods = 1 to GenHdr.QUSNBRLE;                              

         if ((SrvPgmE.QBNBMN00 = ModNam) and                          

            ((SrvPgmE.QBNBMLN00 = ModNamLib) or (ModNamLib ='*ALL')));

            dsply ('*SRVPGM: ' + %trimr(SrvPgmE.QBNSLIBN) +           

                   '/' + SrvPgmE.QBNSN00);                            

            Hits += 1;                                                

            leave;                                                    

         endif;                                                       

         SrvPgmEPtr += GenHdr.QUSSEE;                                 

     endfor;                                                          

  else;                                                               

     dsply ('Unable to access *SRVPGMs.');                            

  endif;                                                              

 /end-free                                                            

pChkSrvPgmMod     e                                                   

 *****************************************************************    

pChkPgmMod        b                                                   

dChkPgmMod        pi                                                  

dGetPgmInfo       pr                  extpgm('QBNLPGMI')              

d QUsrSpcNam                    20    const                           

d Format                         8    const                           

d QPgmNam                       20    const                           

d QUSEC                               likeds(QUSEC)                

d/copy qsysinc/qrpglesrc,qbnlpgmi                                  

dPgmE             ds                  likeds(QBNL0100)             

d                                     based(PgmEPtr)               

dPgmEPtr          s               *                                

 /free                                                             

  // Get the list of bound *MODULEs for current *PGM               

  GetPgmInfo( ModLstSpc :'PGML0100'                                

             :(RcvVarE.QGYON00 + RcvVarE.QGYOL03) :ErrCde);        

  select;                                                          

     when ErrCde.Hdr.QUSBAVL = 0;                                  

     when ErrCde.Hdr.QUSEI = 'CPF5CF5';     // Not an ILE *PGM        

          return;                                                     

     other;                                                           

          SndErrMsg();                                                

  endsl;                                                              

  if (GenHdr.QUSIS = 'C');                                            

     PgmEPtr = ModLstSpcPtr + GenHdr.QUSOLD;                          

     for NbrMods = 1 to GenHdr.QUSNBRLE;                              

         if ((PgmE.QBNBMN = ModNam) and                               

            ((PgmE.QBNBMLN = ModNamLib) or (ModNamLib ='*ALL')));     

            dsply ('*PGM: ' + %trimr(PgmE.QBNPGMLN) +                 

                   '/' + PgmE.QBNPGMN00);                             

            Hits += 1;                                                

            leave;                                                    

         endif;                                                       

         PgmEPtr += GenHdr.QUSSEE;                                    

     endfor;                                                         

  else;                                                              

     dsply ('Unable to access *PGMs.');                              

  endif;                                                             

 /end-free                                                           

pChkPgmMod        e                                                  

 *****************************************************************   

pOpnLstDone       b                                                  

dOpnLstDone       pi              n                                  

d RqsHandle                      4                                   

 /free                                                               

  if ((QGYFBR01 + QGYRRTN02) > QGYTR07) and (QGYLSI01 = '2');    

     CloseList( RqsHandle :QUSEC);    // close the open list     

     return *on;                                                 

  else;                                                          

     GetNextEnt( RcvVar :%size(RcvVar) :RqsHandle :QGYLI :50     

                :QGYFBR01 + QGYRRTN02  :ErrCde);                 

     select;                                                     

        when ErrCde.Hdr.QUSBAVL = 0;                             

             return *off;
        when ((ErrCde.Hdr.QUSEI = 'GUI0006') or  

              (ErrCde.Hdr.QUSEI = 'GUI0118'));                                           

             CloseList( RqsHandle :QUSEC);
             return *on;                                         

        other;                                                   

             SndErrMsg();                                        

     endsl;                                                      

  endif;                                                         

 /end-free                                                          

pOpnLstDone       e                                                 

 *****************************************************************  

pSetUp            b                                                 

dSetUp            pi                                                

dCrtUsrSpc        pr                  extpgm('QUSCRTUS')            

d SpcName                       20    const                         

d SpcAttr                       10    const                         

d SpcSiz                        10i 0 const                         

d SpcVal                         1    const                         

d SpcAut                        10    const                         

d SpcTxt                        50    const                         

d SpcRpl                        10    const options(*nopass)          

d QUSEC                               likeds(QUSEC) options(*nopass)  

d SpcDmn                        10    const options(*nopass)          

d SpcTfrSiz                     10i 0 const options(*nopass)          

d SpcSpcAln                      1    const options(*nopass)          

dGetUsrSpcPtr     pr                  extpgm('QUSPTRUS')              

d SpcName                       20    const                           

d UsrSpcPtr                       *                                   

d QUSEC                               likeds(QUSEC) options(*nopass)  

dChgSvrJobs       pr                  extpgm('QGYCHGSJ')              

d NbrJobs                       10i 0 const                           

d QUSEC                               likeds(QUSEC)                   

 /free                                                                

  // set Error code bytes provided to send exceptions                   

  QUSBPRV = 0;                                                          

  // set Error code bytes provided to not send exceptions               

  ErrCde.Hdr.QUSBPRV = %size(ErrCde);                                   

  // Save the module to locate as two subfields                         

  ModToFnd = QModNam;                                                   

  // Set the library to search for *SRVPGMs                             

  ObjLib = SrchLib;                                                     

  // Prime the input fields for the QGYOLOBJ API                        

  QGYNBRK = 0;                         // no need to sort API output    

  QGYOAC = *loval;                     // init authority control struct 

  QGYFL04 = %size(QGYOAC);             // no authority controls needed  

  SelCtl.Ctl = *loval;                 // init selection structure      

  SelCtl.Ctl.QGYFL05 = %size(SelCtl);  // room for one status value    

  SelCtl.Ctl.QGYSOOS = 0;              // select based on status       

  SelCtl.Ctl.QGYSO01 =                 // displacement to first status 

     (%addr(SelCtl) - %addr(SelCtl.Status));                           

  SelCtl.Ctl.QGYNBRS = 1;              // one status value supplied    

  SelCtl.Status = '*';                 // return all entries           

  // Allow up to two concurrent server jobs                            

  ChgSvrJobs(2 :QUSEC);                                                

  // Start the asynchronous creation of a *SRVPGM list                 

  OpnListObj( RcvVar :0 :QGYLI :0 :QGYOSI :ObjLibNam                   

             :'*SRVPGM' :QGYOAC :SelCtl :0 :0 :QUSEC);                 

  SrvPgmHdl = QGYRH07;                        
  OLSrvPgmOpn = *on;                               

  // Start the asynchronous creation of a *PGM list                    

  OpnListObj( RcvVar :0 :QGYLI :0 :QGYOSI :ObjLibNam                   

             :'*PGM' :QGYOAC :SelCtl :0 :0 :QUSEC);                     

  PgmHdl = QGYRH07;                              
  OLPgmOpn = *on;                           

  // Create *USRSPC for module listing                                  

  CrtUsrSpc( ModLstSpc :' ' :1 :x'00' :'*CHANGE' :' ' :'*YES' :QUSEC);  

  // Get pointer to *USRSPC for module listing                          

  GetUsrSpcPtr( ModLstSpc :ModLstSpcPtr :QUSEC);                        

 /end-free                                                              

pSetUp            e                                                     

 *****************************************************************      

pSndErrMsg        b                                                     

dSndErrMsg        pi                                                 

dSndMsg           pr                  extpgm('QSYS/QMHSNDPM')        

d MsgID                          7    const                          

d QualMsgF                      20    const                          

d MsgDta                     65535    const options(*varsize)        

d LenMsgDta                     10i 0 const                          

d MsgType                       10    const                          

d CallStackEntry             65535    const options(*varsize)        

d CallStackCntr                 10i 0 const                          

d MsgKey                         4                                   

d QUSEC                               likeds(QUSEC)                  

d LenCSE                        10i 0 const options(*nopass)         

d CSEQual                       20    const options(*nopass)         

d DSPWaitTime                   10i 0 const options(*nopass)         

d CSEType                       10    const options(*nopass)         

d CCSID                         10i 0 const options(*nopass)         

dQualMsgF         ds                  qualified                      

d MsgFile                       10                                   

d MsgFLib                       10    inz('QSYS')                    

dMsgKey           s              4                                   

 /free                                                               

  if %subst(ErrCde.Hdr.QUSEI :1 :3) = 'GUI';                         

     QualMsgF.MsgFile = 'QGUIMSG';                                   

  else;                                                              

     QualMsgF.MsgFile = 'QCPFMSG';                                   

  endif;                                                             

  if ErrCde.Hdr.QUSBAVL > 16;          // Msg data provided          

     SndMsg( ErrCde.Hdr.QUSEI :QualMsgF :ErrCde.MsgDta               

            :(ErrCde.Hdr.QUSBAVL - 16) :'*ESCAPE' :'*' :0       

            :MsgKey :QUSEC);                                    

  else;                                // No msg data           

     SndMsg( ErrCde.Hdr.QUSEI :QualMsgF :' ' :0                 

            :'*ESCAPE' :'*' :0 :MsgKey :QUSEC);                 

  endif;                                                        

 /end-free                                                      

pSndErrMsg        e                                             
